Q:

Check if cell is within range, if so, return word "Marginal" in Excel VBA

I am attempting to check if the value in column N of an excel spreadsheet is within the range of -10 to 10.  If it is, then return the word "Marginal" in column O, else return the value in N.  I am not sure how to go about doing this.  This is the code that I have so far:
Sub SearchFolders()
'UpdatebySUPERtoolsforExcel2016
    Dim xOut As Worksheet
    Dim xWb As Workbook
    Dim xWks As Worksheet
    Dim InterSectRange As Range

    Application.ScreenUpdating = False
    Set xWb = ActiveWorkbook
    For Each xWks In xWb.Sheets

        xRow = 1
        With xWks
            .Activate                    'Activating the worksheet
            .Cells(xRow, 12) = "Meas-LO"
            .Cells(xRow, 13) = "Meas-Hi"
            .Cells(xRow, 14) = "Min Value"
            .Cells(xRow, 15) = "Marginal"
            LastRow = ActiveSheet.UsedRange.Rows.Count
            Range("L2").Formula = "=G2+I2"
            Range("L2").AutoFill Destination:=Range("L2:L" & LastRow)
            Range("M2").Formula = "=I2-F2"
            Range("M2").AutoFill Destination:=Range("M2:M" & LastRow)
            Range("N2").Formula = "=min(L2,I2)"
            'Set InterSectRange = Application.Intersect(rng1, rng2)
            'If
            Range("N2").AutoFill Destination:=Range("N2:N" & LastRow)
         End With
         Application.ScreenUpdating = True 'turn it back on

     Next xWks
End Sub

A:

For Each xWks In xWb.Sheets

    xRow = 1
    With xWks
        '.Activate                    '<< not required, and best avoided
        .Cells(xRow, 12) = "Meas-LO"
        .Cells(xRow, 13) = "Meas-Hi"
        .Cells(xRow, 14) = "Min Value"
        .Cells(xRow, 15) = "Marginal"
        LastRow = .UsedRange.Rows.Count
        .Range("L2").Formula = "=G2+I2"
        .Range("L2").AutoFill Destination:=.Range("L2:L" & LastRow)
        .Range("M2").Formula = "=I2-F2"
        .Range("M2").AutoFill Destination:=.Range("M2:M" & LastRow)
        .Range("N2").Formula = "=min(L2,I2)"
        .Range("N2").AutoFill Destination:=.Range("N2:N" & LastRow)

        .Range("O2").Formula = "=IF(AND(N2>=-10, N2<=10), ""Marginal"", N2)"
        .Range("O2").AutoFill Destination:=.Range("O2:O" & LastRow)

     End With
     Application.ScreenUpdating = True 'turn it back on

 Next xWks

